
GAMSBAS: A Program for Saving an Advanced Basis for GAMS

A program (GAMSBAS) has been written which saves information providing an advanced
basis for a GAMS model.  This information contains the shadow price, variable levels, and
reduced costs, and is saved in a GAMS readable file.  The file can, in turn, be included in
subsequent GAMS models providing an advanced basis.

General Notes

Use of this procedure is relevant in cases where there are alterations in the data before a
SOLVE statement in a large already solved model.  The model should usually have saved restart
files and the alterations should require rerunning the model from scratch.  The general use of the
procedure requires restarting GAMS after a solve and executing with a procedure which saves the
basis information.  The resultant data will be written on the file *.BAS, where "*" is the name of
the GAMS input file.  This basis then can be included in subsequent runs.  An example of this
procedure is given below.

The basis file should never be used for one time solution of a problem and rarely for
solution of a file without use of restart files.  One should only use this procedure with large
models when one has to manipulate some of the original data sets equations, or variables before
the first solve statement such that the model has to be restarted from scratch.  One might also
wish to preserve a basis from an alternative run.

The implementation of GAMSBAS causes it to go through several steps.  When the
procedure is first called GAMS generates the model and sends it out to the solver.  In turn,
GAMSBAS examines the problem and selects the solver to be used.  Ordinarily, the default solver
for the problem type (whether linear, nonlinear, or integer) is used.  Users may exercise control
over this process by using the options file (GAMSBAS.OPT) as described below.  In turn, the
solver is invoked and then GAMSBAS writes the basis.

During execution the program includes the line $OFFLISTING as the sixth line in the
*.BAS file.  This suppresses the listing of all but the first five lines of the basis in the file that
includes it.  Users wishing the full listing should delete this entry.

GAMS constructs a basis using information from the optimal solution.  This ordinarily
involves the level and marginal value of all variables plus an indicator of whether or not an
equation has a shadow price.  Degeneracies and alternative optimals complicate this process. 
GAMSBAS tries to overcome this by inserting EPS to indicate when a variable is basic or
nonbasic.

Once the GAMSBAS information has been placed into GAMS the basis may not always
be adequate.  For example, a model which took over 100,000 iterations to get an initial solution

required 1200 iterations to reach optimality when restarted from its GAMSBAS basis.  However,
this reflected a considerable time saving.

Program Usage

There are three steps involved in using GAMSBAS.  The first step involves changing the
solver name in the GAMS file.  This is done using the command:

OPTION LP=GAMSBAS 
or

OPTION NLP=GAMSBAS
or

OPTION IP=GAMSBAS 

The solver in this case is named GAMSBAS.

Second, restart the model and generate the basis file.  Let's assume that the model name is
BLOCKDIA.  One would then execute the command GAMS BLOCKDIA with the solve option
inserted before the solve command, as is done in Table 1, line 147.  In turn, the file
BLOCKDIA.BAS is generated.  This file is listed in Table 2.  Note, this file is just a set of GAMS
replacement commands which inserts marginal values for the equations and marginal and level
values for the variables (See the chapter on Basis formation in McCarl et al for an explanation of
GAMS basis formation).

Third, an include command is entered right before the solve in the model to be restarted
and the option selecting GAMSBAS as the solving program is normally eliminated.  This is done
in Table 3 in lines 147-8 (note the OPTION LP=GAMSBAS is commented out).  Use of this
procedure results in the model in Table 1 solving in 0 iterations after inclusion of the basis file as
opposed to 23 iterations before inclusion of the basis file.

One may find that when a basis from one model is included in another model that the
compiler may detect domain errors because the variables are defined over sets with different
structures across the two models.  One can suppress the domain errors by using the GAMS
command $OFFUNI just before inclusion of the basis file and $ONUNI just after.

The OPTION FILE

GAMSBAS internally selects the solver to use.  Users may override this choice by the use
of the options file.  There are keywords allowed in the options file.  These are as follows

OPTION Name Purpose

LP Gives name of solver for LP problem

NLP Gives name of solver for NLP problem

MIP Gives name of solver for MIP problem

DNLP Gives name of solver for DNLP problem

SOLVERNAME Gives name of solver for problem to be used

In each case the option name is followed by the name of one of a licensed solvers.  If the options
file is empty then the default solvers will be used provided it matches the name of a solver GAMS
knows about.

The GAMSBAS option file is called GAMSBAS.OPT.  An example of a file could look
like the following 2 lines

LP OSL
MIP LAMPS

One other important point regarding the option file involves the name of the active solver
options file.  As seen above the GAMSBAS.OPTION file does not include options commands
such as those which should be submitted to MINOS for example.  In all cases the program uses
the default option filename for the particular solver.  Thus if MINOS5 is being used the program
looks for the solver option file on MINOS5.OPT.

Table 1.  Example File

  16  *     SECTION A       SET DEFINITION
  17  
  18   SET  PRODUCT       TABLES CHAIRSSETS    /TABLES, CHAIRS, DINSETS/
  19        TYPE          TYPES OF PRODUCT     /FUNCTIONAL ,FANCY/
  20        RESOURCE      TYPES OF RESOURCES   /SMLLATHE,LRGLATHE,CARVER,LABOR,
                                                                           TOP/
  21        METHOD        PRODUCTION METHODS   /NORMAL,MAXSML,MAXLRG/
  22        PLANT         DIFFERENT PLANTS     /PLANT1, PLANT2, PLANT3/
  23        SUBPRODUCT(PRODUCT)                /TABLES, CHAIRS/;
  24  
  25  *     SECTION B       DATA DEFINITION
  26  
  27   PARAMETER  SETCHAIR(TYPE)  CHAIRS CONTAINED IN EACH SET
  28                            / FUNCTIONAL 4, FANCY 6/
  29             TABLECOST(TYPE)  TABLECOST  /FUNCTIONAL 80,FANCY 100/;
  30  
  31   TABLE  CHAIRCOST(METHOD,TYPE) CHAIR COST FOR DIFFERENT METHOD
  32                                 FUNCTIONAL      FANCY
  33              NORMAL                15            25
  34              MAXSML                16            25.7
  35              MAXLRG                16.5          26.6 ;
  36  
  37   TABLE TB1(RESOURCE,TYPE,METHOD)   USE OF RESOURCES IN CHAIR PRODUCTION
  38  
  39              FUNCTIONAL.NORMAL   FUNCTIONAL.MAXSML   FUNCTIONAL.MAXLRG
  40  SMLLATHE          0.8               1.30                0.20
  41  LRGLATHE          0.5               0.20                1.30
  42  CARVER            0.4               0.40                0.40
  43  LABOR             1.0               1.05                1.10
  44   +
  45  
  46               FANCY.NORMAL       FANCY.MAXSML        FANCY.MAXLRG
  47  SMLLATHE         1.2                1.7                  0.50
  48  LRGLATHE         0.7                0.30                 1.50
  49  CARVER           1.0                1.00                 1.00
  50  LABOR            0.8                0.82                 0.84;
  51  
  52   TABLE TB2(RESOURCE,TYPE)     USE OF RESOURCES IN TABLE PRODUCTION
  53  
  54                 FUNCTIONAL     FANCY
  55     LABOR          3            5
  56     TOP            1            1;
  57  
  58   TABLE TRANSCOST(SUBPRODUCT, PLANT,TYPE) TRANSPORT COST TO PLANT1
  59  
  60               PLANT1.FUNCTIONAL     PLANT2.FUNCTIONAL  PLANT3.FUNCTIONAL
  61      CHAIRS                                5               7
  62      TABLES                                                20
  63     +
  64               PLANT1.FANCY     PLANT2.FANCY     PLANT3.FANCY
  65      CHAIRS                         5                 7
  66      TABLES                                          20;
  67  
  68   TABLE PRICE(PRODUCT,TYPE) PRICE OF CHAIRS
  69                    FUNCTIONAL     FANCY
  70          CHAIRS       82            105
  71          TABLES      200            300
  72          DINSETS     600           1100;
  73  
  74   TABLE RESORAVAIL(RESOURCE,PLANT) RESOURCES AVAILABLE
  75                  PLANT1      PLANT2      PLANT3
  76       TOP          50                      40
  77       SMLLATHE                 140        130
  78       LRGLATHE                  90        100
  79       CARVER                   120        110
  80       LABOR        175         125        210;

Table 1.  Example File (Continued)

  82   TABLE ACTIVITY(PRODUCT,PLANT) TELLS IF A PLANT SELLS A PRODUCT
  83              PLANT1   PLANT2   PLANT3
  84    TABLES      1                 1
  85    CHAIRS               1        1
  86    DINSETS     1                    ;
  87  
  88  *   SECTION     C      MODEL  DEFINITION
  89  
  90   POSITIVE VARIABLES
  91           MAKECHAIR(PLANT, TYPE,METHOD)  NUMBER OF CHAIRS MADE
  92           MAKETABLE(PLANT, TYPE)         NUMBER OF TABLES MADE
  93           TRNSPORT(PLANT,SUBPRODUCT,TYPE) NUMBER OF ITEMS TRANSPORTED
  94           SELL(PLANT,PRODUCT,TYPE)        NUMBER OF ITEMS SOLD;
  95  
  96   VARIABLES
  97           NETINCOME    NET REVENUE (PROFIT);
  98   EQUATIONS
  99            OBJT         OBJECTIVE FUNCTION ( NET REVENUE )
 100            RESOUREQ(PLANT,RESOURCE)
 101            LINKTABLE(TYPE)  OVERALL FIRM TABLE LINKAGE CONSTRAINTS
 102            LINKCHAIR(TYPE)  OVERALL FIRM CHAIR LINKAGE CONSTRAINTS
 103            TRNCHAIREQ(PLANT,TYPE) CHAIRS BALANCE FOR A PLANT
 104            TRNTABLEEQ(PLANT,TYPE) TABLES BALANCE FOR A PLANT;
 105  
 106    OBJT..      NETINCOME =E=
 107        SUM((TYPE, PRODUCT, PLANT)$ACTIVITY(PRODUCT,PLANT),
 108                   PRICE(PRODUCT,TYPE) * SELL(PLANT,PRODUCT,TYPE))
 109      - SUM((PLANT,TYPE)$ACTIVITY("TABLES",PLANT),
 110               MAKETABLE(PLANT,TYPE)*TABLECOST(TYPE))
 111      - SUM((PLANT,TYPE,METHOD)$ACTIVITY("CHAIRS",PLANT),
 112               MAKECHAIR(PLANT,TYPE,METHOD)* CHAIRCOST(METHOD,TYPE))
 113      - SUM((PLANT,TYPE,SUBPRODUCT)$TRANSCOST(SUBPRODUCT,PLANT,TYPE),
 114             TRANSCOST(SUBPRODUCT,PLANT,TYPE) * TRNSPORT(PLANT,SUBPRODUCT,
 115  
 116    RESOUREQ(PLANT,RESOURCE)..
 117          SUM((TYPE,METHOD)$ACTIVITY("CHAIRS",PLANT), TB1(RESOURCE, TYPE,METHOD)
 118               * MAKECHAIR(PLANT,TYPE,METHOD)) + SUM(TYPE$TB2(RESOURCE, TYPE),
 119               TB2(RESOURCE,TYPE) * MAKETABLE(PLANT,TYPE))
 120               =L= RESORAVAIL(RESOURCE,PLANT) ;
 121  
 122    LINKTABLE(TYPE)..
 123              SUM(PRODUCT$ACTIVITY(PRODUCT,"PLANT1"), SELL("PLANT1", PRODUCT,TYPE))
 124              =L= MAKETABLE("PLANT1",TYPE) +
 125                  SUM(PLANT$TRANSCOST("TABLES",PLANT,TYPE),
 126                      TRNSPORT(PLANT,"TABLES",TYPE));
 127  
 128    LINKCHAIR(TYPE)..
 129               SELL("PLANT1","DINSETS",TYPE) * SETCHAIR(TYPE)
 130               =L= SUM(PLANT$TRANSCOST("CHAIRS",PLANT,TYPE),
 131                   TRNSPORT(PLANT,"CHAIRS",TYPE));
 132  
 133    TRNCHAIREQ(PLANT,TYPE)..
 134               (TRNSPORT(PLANT,"CHAIRS",TYPE) + SELL(PLANT,"CHAIRS",TYPE))
 135                $TRANSCOST("CHAIRS",PLANT,TYPE)
 136                =L= SUM(METHOD$ACTIVITY("CHAIRS",PLANT),
 137                    MAKECHAIR(PLANT,TYPE,METHOD));
 138  
 139    TRNTABLEEQ(PLANT,TYPE)..
 140               (TRNSPORT(PLANT,"TABLES",TYPE) + SELL(PLANT,"TABLES",TYPE)
 141                - MAKETABLE(PLANT,TYPE))$TRANSCOST("TABLES",PLANT,TYPE)
 142                =L= 0 ;
 143  
 144   MODEL Furn /ALL/;
 145  
 146  * SECTION D    SOLVE THE PROBLEM
 147   option lp=gamsbas
 148   SOLVE Furn USING LP MAXIMIZING NETINCOME; 149  

Table 2.  Basis File

 OBJT.m  =    1.00000000000    ;
 RESOUREQ.m ("PLANT1","LABOR")  =    44.0000000000    ;
 RESOUREQ.m ("PLANT2","SMLLATHE")  =    47.7696526508    ;
 RESOUREQ.m ("PLANT2","LRGLATHE")  =    38.8299817185    ;
 RESOUREQ.m ("PLANT2","LABOR")  =    19.3692870201    ;    
 $ OFFLISTING;
 RESOUREQ.m ("PLANT3","SMLLATHE")  =    18.4975609756    ;
 RESOUREQ.m ("PLANT3","LRGLATHE")  =    12.1853658537    ;
 RESOUREQ.m ("PLANT3","CARVER")  =    35.2731707317    ;
 RESOUREQ.m ("PLANT3","LABOR")  =    40.0000000000    ;
 LINKTABLE.m ("FUNCTIONAL")  =    212.000000000    ;
 LINKTABLE.m ("FANCY")  =    320.000000000    ;
 LINKCHAIR.m ("FUNCTIONAL")  =    97.0000000000    ;
 LINKCHAIR.m ("FANCY")  =    130.000000000    ;
 TRNCHAIREQ.m ("PLANT2","FUNCTIONAL")  =    92.0000000000    ;
 TRNCHAIREQ.m ("PLANT2","FANCY")  =    125.000000000    ;
 TRNCHAIREQ.m ("PLANT3","FUNCTIONAL")  =    90.0000000000    ;
 TRNCHAIREQ.m ("PLANT3","FANCY")  =    123.000000000    ;
 TRNTABLEEQ.m ("PLANT3","FUNCTIONAL")  =    200.000000000    ;
 TRNTABLEEQ.m ("PLANT3","FANCY")  =    300.000000000    ;
 MAKECHAIR.l ("PLANT2","FUNCTIONAL","NORMAL")  =    62.2333942718    ;
 MAKECHAIR.m ("PLANT2","FUNCTIONAL","MAXSML")  =   -14.2042961609    ;
 MAKECHAIR.m ("PLANT2","FUNCTIONAL","MAXLRG")  =   -5.83912248629    ;
 MAKECHAIR.l ("PLANT2","FANCY","NORMAL")  =    73.0195003047    ;
 MAKECHAIR.m ("PLANT2","FANCY","MAXSML")  =   -9.44021937843    ;
 MAKECHAIR.l ("PLANT2","FANCY","MAXLRG")  =    5.17976843388    ;
 MAKECHAIR.l ("PLANT3","FUNCTIONAL","NORMAL")  =    35.3658536585    ;
 MAKECHAIR.m ("PLANT3","FUNCTIONAL","MAXSML")  =   -8.59317073171    ;
 MAKECHAIR.m ("PLANT3","FUNCTIONAL","MAXLRG")  =   -4.14975609756    ;
 MAKECHAIR.l ("PLANT3","FANCY","NORMAL")  =    76.8292682927    ;
 MAKECHAIR.m ("PLANT3","FANCY","MAXSML")  =   -5.87463414634    ;
 MAKECHAIR.l ("PLANT3","FANCY","MAXLRG")  =    19.0243902439    ;
 MAKETABLE.l ("PLANT1","FUNCTIONAL")  =    24.3998119826    ;
 MAKETABLE.l ("PLANT1","FANCY")  =    20.3601128105    ;
 MAKETABLE.m ("PLANT2","FUNCTIONAL")  =   -58.1078610603    ;
 MAKETABLE.m ("PLANT2","FANCY")  =   -96.8464351005    ;
 MAKETABLE.l ("PLANT3","FANCY")  =    19.4380487805    ;
 MAKETABLE.m ("PLANT3", "FUNCTIONAL") = EPS;
 TRNSPORT.l ("PLANT2","CHAIRS","FUNCTIONAL")  =    62.2333942718    ;
 TRNSPORT.l ("PLANT2","CHAIRS","FANCY")  =    78.1992687386    ;
 TRNSPORT.m ("PLANT3","TABLES","FUNCTIONAL")  =   -8.00000000000    ;
 TRNSPORT.l ("PLANT3","TABLES","FANCY")  =    8.64870840207    ;
 TRNSPORT.l ("PLANT3","CHAIRS","FUNCTIONAL")  =    35.3658536585    ;
 TRNSPORT.l ("PLANT3","CHAIRS","FANCY")  =    95.8536585366    ;
 SELL.m ("PLANT1","TABLES","FUNCTIONAL")  =   -12.0000000000    ;
 SELL.m ("PLANT1","TABLES","FANCY")  =   -20.0000000000    ;
 SELL.l ("PLANT1","DINSETS","FUNCTIONAL")  =    24.3998119826    ;
 SELL.l ("PLANT1","DINSETS","FANCY")  =    29.0088212125    ;
 SELL.m ("PLANT2","CHAIRS","FUNCTIONAL")  =  -10.00000000000    ;
 SELL.m ("PLANT2","CHAIRS","FANCY")  =   -20.0000000000    ;
 SELL.l ("PLANT3","TABLES","FANCY")  =    10.7893403784    ;
 SELL.m ("PLANT3","CHAIRS","FUNCTIONAL")  =   -8.00000000000    ;
 SELL.m ("PLANT3","CHAIRS","FANCY")  =   -18.0000000000    ;
 NETINCOME.l  =    36206.8788960    ;

Table 3.  Example with Basis File Included

  16  *     SECTION A       SET DEFINITION
  17  
  18   SET  PRODUCT     TABLES CHAIRSSETS    /TABLES, CHAIRS, DINSETS/
  19        TYPE        TYPES OF PRODUCT     /FUNCTIONAL ,FANCY/
  20        RESOURCE    TYPES OF RESOURCES  /SMLLATHE,LRGLATHE,CARVER,LABOR, TOP/
  21        METHOD        PRODUCTION METHODS   /NORMAL,MAXSML,MAXLRG/
  22        PLANT         DIFFERENT PLANTS     /PLANT1, PLANT2, PLANT3/
  23        SUBPRODUCT(PRODUCT)                /TABLES, CHAIRS/;
  24  
  25  *     SECTION B       DATA DEFINITION
  26  
  27   PARAMETER  SETCHAIR(TYPE)  CHAIRS CONTAINED IN EACH SET
  28                            / FUNCTIONAL 4, FANCY 6/
  29             TABLECOST(TYPE)  TABLECOST  /FUNCTIONAL 80,FANCY 100/;
  30  
  31   TABLE  CHAIRCOST(METHOD,TYPE) CHAIR COST FOR DIFFERENT METHOD
  32                                 FUNCTIONAL      FANCY
  33              NORMAL                15            25
  34              MAXSML                16            25.7
  35              MAXLRG                16.5          26.6 ;
  36  
  37   TABLE TB1(RESOURCE,TYPE,METHOD)   USE OF RESOURCES IN CHAIR PRODUCTION
  38  
  39              FUNCTIONAL.NORMAL   FUNCTIONAL.MAXSML   FUNCTIONAL.MAXLRG
  40  SMLLATHE          0.8               1.30                0.20
  41  LRGLATHE          0.5               0.20                1.30
  42  CARVER            0.4               0.40                0.40
  43  LABOR             1.0               1.05                1.10
  44   +
  45  
  46               FANCY.NORMAL       FANCY.MAXSML        FANCY.MAXLRG
  47  SMLLATHE         1.2                1.7                  0.50
  48  LRGLATHE         0.7                0.30                 1.50
  49  CARVER           1.0                1.00                 1.00
  50  LABOR            0.8                0.82                 0.84;
  51  
  52   TABLE TB2(RESOURCE,TYPE)     USE OF RESOURCES IN TABLE PRODUCTION
  53  
  54                 FUNCTIONAL     FANCY
  55     LABOR          3            5
  56     TOP            1            1;
  57  
  58   TABLE TRANSCOST(SUBPRODUCT, PLANT,TYPE) TRANSPORT COST TO PLANT1
  60               PLANT1.FUNCTIONAL     PLANT2.FUNCTIONAL  PLANT3.FUNCTIONAL
  61      CHAIRS                                5               7
  62      TABLES                                                20
  63     +
  64               PLANT1.FANCY     PLANT2.FANCY     PLANT3.FANCY
  65      CHAIRS                         5                 7
  66      TABLES                                          20;
  67  
  68   TABLE PRICE(PRODUCT,TYPE) PRICE OF CHAIRS
  69                    FUNCTIONAL     FANCY
  70          CHAIRS       82            105
  71          TABLES      200            300
  72          DINSETS     600           1100;
  73  
  74   TABLE RESORAVAIL(RESOURCE,PLANT) RESOURCES AVAILABLE
  75                  PLANT1      PLANT2      PLANT3
  76       TOP          50                      40
  77       SMLLATHE                 140        130
  78       LRGLATHE                  90        100
  79       CARVER                   120        110
  80       LABOR        175         125        210;
  81  
  82   TABLE ACTIVITY(PRODUCT,PLANT) TELLS IF A PLANT SELLS A PRODUCT
  83              PLANT1   PLANT2   PLANT3

Table 3.  Example with Basis File Included (Continued)

  84    TABLES      1                 1
  85    CHAIRS               1        1
  86    DINSETS     1                    ;
  87  
  88  *   SECTION     C      MODEL  DEFINITION
  89  
  90   POSITIVE VARIABLES
  91           MAKECHAIR(PLANT, TYPE,METHOD)  NUMBER OF CHAIRS MADE
  92           MAKETABLE(PLANT, TYPE)         NUMBER OF TABLES MADE
  93           TRNSPORT(PLANT,SUBPRODUCT,TYPE) NUMBER OF ITEMS TRANSPORTED
  94           SELL(PLANT,PRODUCT,TYPE)        NUMBER OF ITEMS SOLD;
  95  
  96   VARIABLES
  97           NETINCOME    NET REVENUE (PROFIT);
  98   EQUATIONS
  99            OBJT         OBJECTIVE FUNCTION ( NET REVENUE )
 100            RESOUREQ(PLANT,RESOURCE)
 101            LINKTABLE(TYPE)  OVERALL FIRM TABLE LINKAGE CONSTRAINTS
 102            LINKCHAIR(TYPE)  OVERALL FIRM CHAIR LINKAGE CONSTRAINTS
 103            TRNCHAIREQ(PLANT,TYPE) CHAIRS BALANCE FOR A PLANT
 104            TRNTABLEEQ(PLANT,TYPE) TABLES BALANCE FOR A PLANT;
 105  
 106    OBJT..      NETINCOME =E=
 107        SUM((TYPE, PRODUCT, PLANT)$ACTIVITY(PRODUCT,PLANT),
 108                   PRICE(PRODUCT,TYPE) * SELL(PLANT,PRODUCT,TYPE))
 109      - SUM((PLANT,TYPE)$ACTIVITY("TABLES",PLANT),
 110               MAKETABLE(PLANT,TYPE)*TABLECOST(TYPE))
 111      - SUM((PLANT,TYPE,METHOD)$ACTIVITY("CHAIRS",PLANT),
 112               MAKECHAIR(PLANT,TYPE,METHOD)* CHAIRCOST(METHOD,TYPE))
 113      - SUM((PLANT,TYPE,SUBPRODUCT)$TRANSCOST(SUBPRODUCT,PLANT,TYPE),
 114             TRANSCOST(SUBPRODUCT,PLANT,TYPE) * TRNSPORT(PLANT,SUBPRODUCT, TYPE));
 115  
 116    RESOUREQ(PLANT,RESOURCE)..
 117               SUM((TYPE,METHOD)$ACTIVITY("CHAIRS",PLANT), TB1(RESOURCE,TYPE,METHOD)
 118               * MAKECHAIR(PLANT,TYPE,METHOD)) + SUM(TYPE$TB2(RESOURCE, TYPE),
 119               TB2(RESOURCE,TYPE) * MAKETABLE(PLANT,TYPE))
 120               =L= RESORAVAIL(RESOURCE,PLANT) ;
 121  
 122    LINKTABLE(TYPE)..
 123              SUM(PRODUCT$ACTIVITY(PRODUCT,"PLANT1"), SELL("PLANT1",PRODUCT,TYPE))
 124              =L= MAKETABLE("PLANT1",TYPE) +
 125                  SUM(PLANT$TRANSCOST("TABLES",PLANT,TYPE),
 126                      TRNSPORT(PLANT,"TABLES",TYPE));
 127  
 128    LINKCHAIR(TYPE)..
 129               SELL("PLANT1","DINSETS",TYPE) * SETCHAIR(TYPE)
 130               =L= SUM(PLANT$TRANSCOST("CHAIRS",PLANT,TYPE),
 131                   TRNSPORT(PLANT,"CHAIRS",TYPE));
 132  
 133    TRNCHAIREQ(PLANT,TYPE)..
 134               (TRNSPORT(PLANT,"CHAIRS",TYPE) + SELL(PLANT,"CHAIRS",TYPE))
 135                $TRANSCOST("CHAIRS",PLANT,TYPE)
 136                =L= SUM(METHOD$ACTIVITY("CHAIRS",PLANT),
 137                    MAKECHAIR(PLANT,TYPE,METHOD));
 138  
 139    TRNTABLEEQ(PLANT,TYPE)..
 140               (TRNSPORT(PLANT,"TABLES",TYPE) + SELL(PLANT,"TABLES",TYPE)
 141                - MAKETABLE(PLANT,TYPE))$TRANSCOST("TABLES",PLANT,TYPE)
 142                =L= 0 ;
 143  
 144   MODEL Furn /ALL/;
 145  
 146  * SECTION D    SOLVE THE PROBLEM
 147  * option lp=gamsbas
 148  $INCLUDE  "blockdia.bas"
 149  SOLVE Furn USING LP MAXIMIZING NETINCOME;


